



AN EXPLICIT RECURSIVE FORMULA FOR COMPUTING THE NORMAL FORM AND CENTER MANIFOLD OF GENERAL n -DIMENSIONAL DIFFERENTIAL SYSTEMS ASSOCIATED WITH HOPF BIFURCATION*

YUN TIAN[†] and PEI YU[‡]
*Department of Applied Mathematics,
Western University, London,
Ontario N6A 5B7, Canada*
[†]*ytian56@uwo.ca*
[‡]*pyu@uwo.ca*

Received August 16, 2012

An explicit, computationally efficient, recursive formula is presented for computing the normal form and center manifold of general n -dimensional systems associated with Hopf bifurcation. Maple program is developed based on the analytical formulas, and shown to be computationally efficient, using two examples.

Keywords: Differential systems; Hopf bifurcation; normal forms; center manifold; computational efficiency; symbolic computation; Maple.

1. Introduction

The second part of Hilbert's 16th problem is to decide an upper bound for the number of limit cycles in a planar polynomial vector field of degree n ; it is very complicated. The particular version of this problem is to estimate the number $M(n)$ of small limit cycles bifurcating from a singular point; it is still very difficult. Only for the quadratic case, Bautin [1952] proved $M(2) = 3$. For $n > 2$ this problem is still open. More recent new progress can be found in [Li, 2003].

For general n -dimensional systems, two of the useful tools in the study of stability and bifurcations near singular points are center manifold theory and normal form theory. The center manifold theory can be applied to reduce the dimension of the state

spaces which need to be considered when some eigenvalues of the linearization have zero real part. The basic idea of normal form theory is to transform the original system to a simpler one which keeps the topological structure of the original system around the singular point. Most developments in this direction for the past three decades can be found in [Guckenheimer & Holmes, 1993; Nayfeh, 1993; Chow *et al.*, 1994].

Since computation of normal forms is very involved and time consuming, in particular, for higher-order normal forms, computer algebra systems such as Maple, Mathematica, must be used. Several efficient methodologies for computing normal forms have been developed in the past decade (e.g. see [Yu, 1998, 2003; Giné & Santallusia, 2001]).

*This work was supported by the Natural Science and Engineering Research Council of Canada (NSERC).

[†]Author for correspondence

Recently, researchers have also paid attention to computation of the simplest normal forms (e.g. see [Algaba *et al.*, 1998; Yu, 1999; Yu & Yuan, 2003; Yu & Leung, 2003a; Gazor & Yu, 2012]).

The method of multiple time scales combined with a perturbation technique can be applied to obtain unique normal forms of differential equations for a number of different singularities such as Hopf [Yu, 1998], Hopf-zero [Yu & Leung, 2003b], double Hopf [Yu, 2001, 2002], etc. This method does not need to solve differential equations, nor involve integration, but only need algebraic calculations, which greatly facilitates implementation using computer algebra systems such as Maple. For Hopf bifurcation, this method only requires solving two-dimensional matrix systems for any higher-order normal forms of general n -dimensional systems.

The method of Poincaré–Lyapunov constants can be used to study bifurcations of small limit cycles. Lloyd and Pearson [1990] used computer algebra system Reduce to compute Poincaré–Lyapunov constants by solving linear equations. Giné and Santallusia [2001] obtained a recursive formula for the Poincaré–Lyapunov constants of Hopf bifurcation for general two-dimensional systems, which can be computed recursively in terms of the coefficients of the original system (for its applications, e.g. see [Giné & Santallusia, 2004; Giné & Mallol, 2009]).

Yu [2003] computed the center manifold of differential equations with a proper nonlinear transformation which is incorporated with normal form computation to develop a unified procedure for computing normal forms of general n -dimensional systems. The formulas developed in [Yu, 1998, 2003] are in recursive format, but not explicit, which may involve some repetitive computations, and so may demand more memory in a computer to obtain higher-order normal forms or focus values. Since practical problems often have Hopf bifurcation in high-dimensional systems, and thus the recursive formula in [Giné & Santallusia, 2001], which only computes focus values, cannot be directly applied to such systems. Moreover, for Hopf bifurcation, one may also need to determine the frequency of motion, implying that normal form, rather than just focus values, need to be computed.

In this paper, based on the result of [Yu, 2003], we will develop an efficient method to compute the normal form for Hopf bifurcation in general n -dimensional dynamical systems. We shall present

explicit recursive formulas for simultaneously computing the center manifold and normal form of a given general system, which is available for the first time in the literature.

2. Main Result

Consider a system of differential equations of the form,

$$\dot{\mathbf{y}} = A\mathbf{y} + \mathbf{G}(\mathbf{y}), \quad \mathbf{y} \in R^n, \quad \mathbf{G}(\mathbf{y}) : R^n \rightarrow R^n, \tag{1}$$

where $\mathbf{G}(\mathbf{0}) = \mathbf{0}$, $D_{\mathbf{y}}\mathbf{G}(\mathbf{0}) = \mathbf{0}$, and it is assumed, without loss of generality, that the matrix A has eigenvalues $i, -i, \lambda_1, \dots, \lambda_{k_1}, \lambda_{k_1+1}, \bar{\lambda}_{k_1+1}, \dots, \lambda_{k_1+k_2}, \bar{\lambda}_{k_1+k_2}$. Here $\lambda_1, \dots, \lambda_{k_1}$ are nonzero real numbers, and $\lambda_{k_1+1}, \dots, \lambda_{k_1+k_2}$ are complex numbers with nonzero real part, and $2 + k_1 + 2k_2 = n$.

Then, there exists a linear transformation,

$$\mathbf{y} = T\mathbf{x},$$

such that (1) can be transformed into

$$\dot{\mathbf{x}} = J\mathbf{x} + \mathbf{f}(\mathbf{x}), \tag{2}$$

with $x_2 = \bar{x}_1$, where

$$\begin{aligned} J &= \text{diag}(i, -i, \lambda_1, \dots, \lambda_{k_1+k_2}, \\ &\quad \bar{\lambda}_{k_1+1}, \dots, \bar{\lambda}_{k_1+k_2}), \\ \mathbf{f}(\mathbf{x}) &= \sum_{m \geq 2} \mathbf{f}_m(\mathbf{x}) \\ &= \sum_{m \geq 2} \sum_{\tilde{m}} \mathbf{a}_{\tilde{m}} x_1^{m_1} x_2^{m_2} \cdots x_n^{m_n}, \end{aligned}$$

and $\tilde{m} \triangleq m_1 m_2 \cdots m_n$, denoting a choice of values of m_1, m_2, \dots, m_n which satisfies $\sum_{j=1}^n m_j = m$ with $m_j \geq 0$.

Let $\mathbf{x} = (x_1, \bar{x}_1, \mathbf{x}_r^T)^T$ and $J = \text{diag}(i, -i, J_r)$. Then, Eq. (2) can be written as

$$\begin{aligned} \dot{x}_1 &= ix_1 + f_1(x_1, \bar{x}_1, \mathbf{x}_r), \\ \dot{\bar{x}}_1 &= -i\bar{x}_1 + \bar{f}_1(x_1, \bar{x}_1, \mathbf{x}_r), \\ \dot{\mathbf{x}}_r &= J_r \mathbf{x}_r + \mathbf{f}_r(x_1, \bar{x}_1, \mathbf{x}_r). \end{aligned} \tag{3}$$

Note that the second equation of (3) is a complex conjugate of the first equation.

The center manifold of (3) may be defined in the form of

$$\begin{aligned} \mathbf{x}_r &= \mathbf{H}(x_1, \bar{x}_1) \\ &= \sum_{m \geq 2} \mathbf{H}_m(x_1, \bar{x}_1) \quad \text{with} \\ \mathbf{H}_m(x_1, \bar{x}_1) &= \sum_{j=0}^m \mathbf{h}_j^m x_1^j \bar{x}_1^{m-j}, \end{aligned} \quad (4)$$

which satisfies

$$\begin{aligned} &\mathbf{H}_{x_1}(x_1, \bar{x}_1)\dot{x}_1 + \mathbf{H}_{\bar{x}_1}(x_1, \bar{x}_1)\dot{\bar{x}}_1 \\ &= J_r \mathbf{H}(x_1, \bar{x}_1) + \mathbf{f}_r(x_1, \bar{x}_1, \mathbf{H}(x_1, \bar{x}_1)). \end{aligned}$$

Then, the differential equations describing the dynamics on the center manifold are given by

$$\begin{aligned} \dot{x}_1 &= ix_1 + f_1(x_1, \bar{x}_1, \mathbf{H}(x_1, \bar{x}_1)), \\ \dot{\bar{x}}_1 &= -i\bar{x}_1 + \bar{f}_1(x_1, \bar{x}_1, \mathbf{H}(x_1, \bar{x}_1)). \end{aligned} \quad (5)$$

Next, introduce the transformation, given by

$$\begin{aligned} x_1 &= u + Q(u, \bar{u}) \\ &= u + \sum_{m \geq 2} Q_m(u, \bar{u}) \\ &\equiv q(u, \bar{u}) \end{aligned}$$

with $Q_m(u, \bar{u}) = \sum_{j=0}^m q_j^m u^j \bar{u}^{m-j}$, into (5) to obtain the normal form,

$$\begin{aligned} \dot{u} &= iu + C(u, \bar{u}) \quad \text{where} \\ C(u, \bar{u}) &= \sum_{m \geq 1} a_m u^{m+1} \bar{u}^m. \end{aligned} \quad (6)$$

Let $\mathbf{h}(u, \bar{u}) = \mathbf{H}(q(u, \bar{u}), \bar{q}(u, \bar{u}))$ and

$$\begin{aligned} F_1(u, \bar{u}) &= f_1(q(u, \bar{u}), \bar{q}(u, \bar{u}), \mathbf{h}(u, \bar{u})), \\ \mathbf{F}_r(u, \bar{u}) &= \mathbf{f}_r(q(u, \bar{u}), \bar{q}(u, \bar{u}), \mathbf{h}(u, \bar{u})). \end{aligned}$$

Then we have the following equations

$$\begin{aligned} &\begin{pmatrix} Q_u(u, \bar{u}) & Q_{\bar{u}}(u, \bar{u}) \\ \mathbf{h}_u(u, \bar{u}) & \mathbf{h}_{\bar{u}}(u, \bar{u}) \end{pmatrix} \begin{pmatrix} iu \\ -i\bar{u} \end{pmatrix} - \begin{pmatrix} iQ(u, \bar{u}) \\ J_r \mathbf{h}(u, \bar{u}) \end{pmatrix} \\ &= \begin{pmatrix} F_1(u, \bar{u}) \\ \mathbf{F}_r(u, \bar{u}) \end{pmatrix} - \begin{pmatrix} Q_u(u, \bar{u}) & Q_{\bar{u}}(u, \bar{u}) \\ \mathbf{h}_u(u, \bar{u}) & \mathbf{h}_{\bar{u}}(u, \bar{u}) \end{pmatrix} \\ &\quad \times \begin{pmatrix} C(u, \bar{u}) \\ \bar{C}(u, \bar{u}) \end{pmatrix} - \begin{pmatrix} C(u, \bar{u}) \\ 0 \end{pmatrix}. \end{aligned} \quad (7)$$

Solving (7) order by order, we obtain the center manifold and the normal form as well as the associated nonlinear transformation.

Suppose for $k \geq 0$,

$$\begin{aligned} q^k(u, \bar{u}) &= \sum_{m=k}^{\infty} \sum_{j=0}^m q_j^{k,m} u^j \bar{u}^{m-j}, \\ \mathbf{h}^k(u, \bar{u}) &= \sum_{m=2k}^{\infty} \sum_{j=0}^m \tilde{\mathbf{h}}_j^{k,m} u^j \bar{u}^{m-j}. \end{aligned} \quad (8)$$

We have the following result.

Theorem 1. For the differential system (3), the recursive formulas for the coefficients of the center manifold (4) and the normal form (6) are given as follows: for $s \geq 2$, $0 \leq j \leq s$,

(1) if s is even, then

$$i(2j - s - 1)q_j^s = a_j^s - C_{1,j}^s,$$

$$i(2j - s)\tilde{\mathbf{h}}_j^s - J_r \tilde{\mathbf{h}}_j^s = \mathbf{b}_j^s - \mathbf{C}_{r,j}^s;$$

(2) if s is odd, then

$$i(2j - s - 1)q_j^s = a_j^s - C_{1,j}^s, \quad \text{for } j \neq \frac{s+1}{2},$$

$$a_{\frac{s-1}{2}}^s = a_{\frac{s+1}{2}}^s - C_{1, \frac{s+1}{2}}^s,$$

$$i(2j - s)\tilde{\mathbf{h}}_j^s - J_r \tilde{\mathbf{h}}_j^s = \mathbf{b}_j^s - \mathbf{C}_{r,j}^s;$$

where $\tilde{\mathbf{h}}_j^{m,s} = (\tilde{h}_{j,1}^{m,s}, \tilde{h}_{j,2}^{m,s}, \dots, \tilde{h}_{j,n-2}^{m,s})^T$ and $\mathbf{a}_j^s = (a_j^s, \bar{a}_j^s, \mathbf{b}_j^{sT})^T$,

$$\begin{aligned} \mathbf{a}_j^s &= \sum_{m=2}^s \sum_{\bar{m}} \sum_{\bar{s}} \sum_{\bar{j}} \mathbf{a}_{\bar{m}} q_{j_1}^{m_1, s_1} \bar{q}_{j_2}^{m_2, s_2} \tilde{h}_{j_3, 1}^{m_3, s_3} \tilde{h}_{j_4, 2}^{m_4, s_4} \dots \tilde{h}_{j_{n-1}, n-3}^{m_{n-1}, s_{n-1}} \tilde{h}_{j_n, n-2}^{m_n, s_n} \\ \tilde{\mathbf{h}}_j^s &= \mathbf{h}_j^s + \sum_{m=2}^{s-1} \sum_{k=0}^m \sum_{l_1=k}^{s+k-m} \sum_{j_1=\max(0, l_1+j-s)}^{\min(l_1, j)} \mathbf{h}_k^m q_{j_1}^{k, l_1} \bar{q}_{s+j_1-l_1-j}^{m-k, s-l_1} \end{aligned}$$

$$C_{1,j}^s = \sum_{m=1}^{\min([\frac{s-2}{2}], j, s-j)} ((j-m)a_m + (s-j-m)\bar{a}_m)q_{j-m}^{s-2m},$$

$$C_{r,j}^s = \sum_{m=1}^{\min([\frac{s-2}{2}], j, s-j)} ((j-m)a_m + (s-j-m)\bar{a}_m)\tilde{\mathbf{h}}_{j-m}^{s-2m},$$

$$q_j^{m,s} = \sum_{l=m-1}^{s-1} \sum_{j_1=\max(0, l+j-s)}^{\min(l,j)} q_{j_1}^{m-1,l} q_{j-j_1}^{s-l}, \quad m \geq 2,$$

$$\tilde{\mathbf{h}}_j^{m,s} = \sum_{l=m-1}^{s-1} \sum_{j_1=\max(0, l+j-s)}^{\min(l,j)} \tilde{\mathbf{h}}_{j_1}^{m-1,l} \tilde{\mathbf{h}}_{j-j_1}^{s-l}, \quad m \geq 2.$$

Proof. For any given integer $s \geq 2$, suppose that we have obtained q_j^n and \mathbf{h}_j^n for $n < s$, $0 \leq j \leq n$ and $a_m, m \leq [\frac{s-2}{2}]$. Now, we want to derive the formulas for q_j^s and \mathbf{h}_j^s for $0 \leq j \leq s$ and $a_{[\frac{s-1}{2}]}$. We divide the proof in three steps, which can also be served as the guidelines for developing programs using a computer algebra system.

Step 1. Denote

$$\mathbf{H}(x_1, \bar{x}_1) = \mathbf{h}(u, \bar{u})$$

$$= \sum_{m=2}^s \sum_{k=0}^m \tilde{\mathbf{h}}_k^m u^k \bar{u}^{m-k} + o(|u, \bar{u}|^s).$$

In this step, we derive the formula for $\tilde{\mathbf{h}}_k^s, 0 \leq k \leq s$. First of all, we need to compute $x_1^k = q^k(u, \bar{u})$, where $2 \leq k \leq s$. Since $q^k(u, \bar{u}) = q(u, \bar{u})q^{k-1}(u, \bar{u})$, we have

$$q^k(u, \bar{u}) = \left(\sum_{m=1}^s \sum_{j=0}^m q_j^m u^j \bar{u}^{m-j} + o(|u, \bar{u}|^s) \right) \left(\sum_{m=k-1}^s \sum_{j=0}^m q_j^{k-1,m} u^j \bar{u}^{m-j} + o(|u, \bar{u}|^s) \right)$$

$$= \sum_{m=k}^s \sum_{j=0}^m \sum_{l=k-1}^{m-1} \sum_{j_1=\max(0, l+j-m)}^{\min(l,j)} q_{j_1}^{k-1,l} q_{j-j_1}^{m-l} u^j \bar{u}^{m-j} + o(|u, \bar{u}|^s).$$

Then, for $2 \leq k \leq s, 0 \leq j \leq s$, we obtain

$$q_j^{k,s} = \sum_{l=k-1}^{s-1} \sum_{j_1=\max(0, l+j-s)}^{\min(l,j)} q_{j_1}^{k-1,l} q_{j-j_1}^{s-l}.$$

For $2 \leq m \leq s$,

$$\mathbf{H}_m(x_1, \bar{x}_1) = \sum_{k=0}^m \mathbf{h}_k^m x_1^k \bar{x}_1^{m-k}$$

$$= \sum_{k=0}^m \mathbf{h}_k^m q^k(u, \bar{u}) \bar{q}^{m-k}(u, \bar{u})$$

$$= \sum_{k=0}^m \mathbf{h}_k^m \left(\sum_{l=k}^s \sum_{j=0}^l q_j^{k,l} u^j \bar{u}^{l-j} + o(|u, \bar{u}|^s) \right) \left(\sum_{l=m-k}^s \sum_{j=0}^l \bar{q}_j^{m-k,l} \bar{u}^j u^{l-j} + o(|u, \bar{u}|^s) \right)$$

$$\begin{aligned}
 &= \sum_{k=0}^m \mathbf{h}_k^m \left(\sum_{l=m}^s \sum_{j=0}^l \sum_{l_1=k}^{l+k-m} \sum_{j_1=\max(0, l_1+j-l)}^{\min(l_1, j)} q_{j_1}^{k, l_1} \bar{q}_{l+j_1-l_1-j}^{m-k, l-l_1} u^j \bar{u}^{l-j} + o(|u, \bar{u}|^s) \right) \\
 &= \sum_{l=m}^s \sum_{j=0}^l \sum_{k=0}^m \sum_{l_1=k}^{l+k-m} \sum_{j_1=\max(0, l_1+j-l)}^{\min(l_1, j)} \mathbf{h}_k^m q_{j_1}^{k, l_1} \bar{q}_{l+j_1-l_1-j}^{m-k, l-l_1} u^j \bar{u}^{l-j} + o(|u, \bar{u}|^s),
 \end{aligned}$$

where $q_j^{0,0} = 1$ and $q_j^{0,l} = 0$ if $l \geq 1$. In particular,

$$\mathbf{H}_s(x_1, \bar{x}_1) = \sum_{j=0}^s \mathbf{h}_j^s u^j \bar{u}^{s-j} + o(|u, \bar{u}|^s).$$

Since $\mathbf{h}(u, \bar{u}) = \mathbf{H}(x_1, \bar{x}_1) = \sum_{m \geq 2} \mathbf{H}_m(x_1, \bar{x}_1)$ and $\mathbf{H}_m(q(u, \bar{u}), \bar{q}(u, \bar{u})) = O(|u, \bar{u}|^m)$, we obtain

$$\tilde{\mathbf{h}}_j^s = \mathbf{h}_j^s + \sum_{m=2}^{s-1} \sum_{k=0}^m \sum_{l_1=k}^{s+k-m} \sum_{j_1=\max(0, l_1+j-s)}^{\min(l_1, j)} \mathbf{h}_k^m q_{j_1}^{k, l_1} \bar{q}_{s+j_1-l_1-j}^{m-k, s-l_1}.$$

Step 2. Denote $\mathbf{F}(u, \bar{u}) = (F_1(u, \bar{u}), \bar{F}_1(u, \bar{u}), \mathbf{F}_r^T(u, \bar{u}))^T$,

$$\mathbf{F}(u, \bar{u}) = \sum_{m=2}^s \sum_{j=0}^m \mathbf{a}_j^m u^j \bar{u}^{m-j} + o(|u, \bar{u}|^s). \tag{9}$$

In this step, we derive the formula for \mathbf{a}_j^s , $0 \leq j \leq s$.

Let $\mathbf{h}^k(u, \bar{u}) = \sum_{m=2k}^s \sum_{j=0}^m \tilde{\mathbf{h}}_j^{k,m} u^j \bar{u}^{m-j} + o(|u, \bar{u}|^s)$, $k \geq 0$, where $\tilde{\mathbf{h}}_j^{1,m} = \tilde{\mathbf{h}}_j^m$, $\tilde{\mathbf{h}}_0^{0,0} = 1$ and $\tilde{\mathbf{h}}_j^{0,m} = 0$ if $m \geq 1$. Using the same method for computing $q_j^{k,s}$, $s \geq 2k$, $0 \leq j \leq s$, we have

$$\tilde{\mathbf{h}}_j^{k,s} = \sum_{l=2k-2}^{s-2} \sum_{j_1=\max(0, j+l-s)}^{\min(l, j)} \tilde{\mathbf{h}}_{j_1}^{k-1, l} \tilde{\mathbf{h}}_{j-j_1}^{s-l}.$$

Let $\tilde{\mathbf{h}}_j^{m,l} = (\tilde{h}_{j,1}^{m,l}, \tilde{h}_{j,2}^{m,l}, \dots, \tilde{h}_{j,n-2}^{m,l})^T$. For $2 \leq m \leq s$, substituting $q(u, \bar{u})$ and $\tilde{\mathbf{h}}(u, \bar{u})$ into $\mathbf{f}_m(\mathbf{x})$ yields

$$\begin{aligned}
 \mathbf{f}_m(\mathbf{x}) &= \sum_{\tilde{m}} \mathbf{a}_{\tilde{m}} x_1^{m_1} x_2^{m_2} \dots x_n^{m_n} \\
 &= \sum_{\tilde{m}} \mathbf{a}_{\tilde{m}} q^{m_1}(u, \bar{u}) \bar{q}^{m_2}(u, \bar{u}) h_1^{m_3}(u, \bar{u}) \dots h_{n-2}^{m_n}(u, \bar{u}) \\
 &= \sum_{\tilde{m}} \mathbf{a}_{\tilde{m}} \left(\sum_{l=m_1}^s \sum_{j=0}^l q_j^{m_1, l} u^j \bar{u}^{l-j} + o(|u, \bar{u}|^s) \right) \left(\sum_{l=m_2}^s \sum_{j=0}^l \bar{q}_j^{m_2, l} \bar{u}^j u^{l-j} + o(|u, \bar{u}|^s) \right) \\
 &\quad \times \left(\sum_{l=2m_3}^s \sum_{j=0}^l \tilde{h}_{j,1}^{m_3, l} u^j \bar{u}^{l-j} + o(|u, \bar{u}|^s) \right) \dots \left(\sum_{l=2m_n}^s \sum_{j=0}^l \tilde{h}_{j, n-2}^{m_n, l} u^j \bar{u}^{l-j} + o(|u, \bar{u}|^s) \right)
 \end{aligned}$$

$$\begin{aligned}
 &= \sum_{\tilde{m}} \mathbf{a}_{\tilde{m}} \left(\sum_{l=m}^s \sum_{j=0}^l \sum_{l_1=m_1}^l \sum_{l_2=m_2}^l \sum_{l_3=2m_3}^l \cdots \sum_{l_n=2m_n}^l \sum_{j_1=0}^{l_1} \sum_{j_2=0}^{l_2} \sum_{j_3=0}^{l_3} \cdots \right. \\
 &\quad \left. \sum_{j_n=0}^{l_n} q_{j_1}^{m_1, l_1} \bar{q}_{j_2}^{m_2, l_2} \tilde{h}_{j_3, 1}^{m_3, l_3} \tilde{h}_{j_4, 2}^{m_4, l_4} \cdots \tilde{h}_{j_{n-1}, n-3}^{m_{n-1}, l_{n-1}} \tilde{h}_{j_n, n-2}^{m_n, l_n} u^j \bar{u}^{l-j} + o(|u, \bar{u}|^s) \right) \\
 &= \sum_{l=m}^s \sum_{j=0}^l \sum_{\tilde{m}} \sum_{\tilde{l}} \sum_{\tilde{j}} \mathbf{a}_{\tilde{m}} q_{j_1}^{m_1, l_1} \bar{q}_{j_2}^{m_2, l_2} \tilde{h}_{j_3, 1}^{m_3, l_3} \tilde{h}_{j_4, 2}^{m_4, l_4} \cdots \tilde{h}_{j_{n-1}, n-3}^{m_{n-1}, l_{n-1}} \tilde{h}_{j_n, n-2}^{m_n, l_n} u^j \bar{u}^{l-j} + o(|u, \bar{u}|^s).
 \end{aligned}$$

Since $\mathbf{F}(u, \bar{u}) = \mathbf{f}(\mathbf{x}) = \sum_{m \geq 2} \mathbf{f}_m(\mathbf{x})$, for $0 \leq j \leq s$, we consequently obtain

$$\mathbf{a}_j^s = \sum_{m=2}^s \sum_{\tilde{m}} \sum_{\tilde{s}} \sum_{\tilde{j}} \mathbf{a}_{\tilde{m}} q_{j_1}^{m_1, s_1} \bar{q}_{j_2}^{m_2, s_2} \tilde{h}_{j_3, 1}^{m_3, s_3} \tilde{h}_{j_4, 2}^{m_4, s_4} \cdots \tilde{h}_{j_{n-1}, n-3}^{m_{n-1}, s_{n-1}} \tilde{h}_{j_n, n-2}^{m_n, s_n},$$

where $0 \leq j_k \leq s_k$, for any $1 \leq k \leq n$, and the index \tilde{s} satisfies that

$$s_k \begin{cases} = 0 & \text{if } m_k = 0, \\ \geq m_k \text{ for } k = 1, 2 \\ \geq 2m_k \text{ for } 3 \leq k \leq n \end{cases} \quad \text{if } m_k \neq 0.$$

Step 3. Denote

$$\begin{pmatrix} Q_u(u, \bar{u}) & Q_{\bar{u}}(u, \bar{u}) \\ \mathbf{h}_u(u, \bar{u}) & \mathbf{h}_{\bar{u}}(u, \bar{u}) \end{pmatrix} \begin{pmatrix} C(u, \bar{u}) \\ \bar{C}(u, \bar{u}) \end{pmatrix} = \begin{pmatrix} \sum_{m=4}^s \sum_{j=1}^m C_{1,j}^m u^j \bar{u}^{m-j} + o(|u, \bar{u}|^s) \\ \sum_{m=4}^s \sum_{j=1}^m \mathbf{C}_{r,j}^m u^j \bar{u}^{m-j} + o(|u, \bar{u}|^s) \end{pmatrix}. \tag{10}$$

In this step, we derive the formulas for $C_{1,j}^s$ and $\mathbf{C}_{r,j}^s$, $0 \leq j \leq s$.

Note that

$$\begin{aligned}
 &Q_u(u, \bar{u})C(u, \bar{u}) + Q_{\bar{u}}(u, \bar{u})\bar{C}(u, \bar{u}) \\
 &= \left(\sum_{m=2}^s \sum_{j=0}^m j q_j^m u^{j-1} \bar{u}^{m-j} + o(|u, \bar{u}|^{s-1}) \right) \left(\sum_{m=1}^{\lfloor \frac{s-1}{2} \rfloor} a_m u^{m+1} \bar{u}^m + o(|u, \bar{u}|^s) \right) \\
 &\quad + \left(\sum_{m=2}^s \sum_{j=0}^m (m-j) q_j^m u^j \bar{u}^{m-j-1} + o(|u, \bar{u}|^{s-1}) \right) \left(\sum_{m=1}^{\lfloor \frac{s-1}{2} \rfloor} \bar{a}_m \bar{u}^{m+1} u^m + o(|u, \bar{u}|^s) \right) \\
 &= \sum_{l=4}^s \sum_{j=2}^l \sum_{m=1}^{\min(\lfloor \frac{l-2}{2} \rfloor, j, l-j)} a_m (j-m) q_{j-m}^{l-2m} u^j \bar{u}^{l-j} + \sum_{l=4}^s \sum_{j=1}^l \sum_{m=1}^{\min(\lfloor \frac{l-2}{2} \rfloor, j, l-j)} \bar{a}_m (l-m-j) q_{j-m}^{l-2m} u^j \bar{u}^{l-j} \\
 &\quad + o(|u, \bar{u}|^s).
 \end{aligned}$$

For the first term in the last expression above, if $j = 1$, then $m = 1$ and $j - m = 0$. So, we obtain

$$Q_u(u, \bar{u})C(u, \bar{u}) + Q_{\bar{u}}(u, \bar{u})\bar{C}(u, \bar{u}) = \sum_{l=4}^s \sum_{j=1}^l \sum_{m=1}^{\min(\lfloor \frac{l-2}{2} \rfloor, j, l-j)} ((j-m)a_m + (l-j-m)\bar{a}_m)q_{j-m}^{l-2m}u^j\bar{u}^{l-j} + o(|u, \bar{u}|^s).$$

Therefore, for $s \geq 4$, $0 \leq j \leq s$, comparing the above equation with (10) we have

$$C_{1,j}^s = \sum_{m=1}^{\min(\lfloor \frac{s-2}{2} \rfloor, j, s-j)} ((j-m)a_m + (s-j-m)\bar{a}_m)q_{j-m}^{s-2m}.$$

Similarly,

$$C_{r,j}^s = \sum_{m=1}^{\min(\lfloor \frac{s-2}{2} \rfloor, j, s-j)} ((j-m)a_m + (s-j-m)\bar{a}_m)\tilde{h}_{j-m}^{s-2m}.$$

Finally, from the left-hand side of (7), we obtain

$$\begin{aligned} & iu\mathbf{h}_u(u \cdot \bar{u}) - i\bar{u}\mathbf{h}_{\bar{u}}(u, \bar{u}) - J_r\mathbf{h}(u, \bar{u}) \\ &= iu \sum_{m=2}^s \sum_{j=0}^m j\tilde{\mathbf{h}}_j^m u^{j-1}\bar{u}^{m-j} - i\bar{u} \sum_{m=2}^s \sum_{j=0}^m (m-j)\tilde{\mathbf{h}}_j^m u^j\bar{u}^{m-j-1} \\ & \quad - J_r \sum_{m=2}^s \sum_{j=0}^m \tilde{\mathbf{h}}_j^m u^j\bar{u}^{m-j} + o(|u, \bar{u}|^s) \\ &= \sum_{m=2}^s \sum_{j=0}^m (i(2j-m)\tilde{\mathbf{h}}_j^m - J_r\tilde{\mathbf{h}}_j^m)u^j\bar{u}^{m-j} + o(|u, \bar{u}|^s), \end{aligned} \tag{11}$$

and similarly,

$$\begin{aligned} & iuQ_u(u \cdot \bar{u}) - i\bar{u}Q_{\bar{u}}(u, \bar{u}) - iQ(u, \bar{u}) \\ &= \sum_{m=2}^s \sum_{j=0}^m i(2j-m-1)q_j^m u^j\bar{u}^{m-j} \\ & \quad + o(|u, \bar{u}|^s). \end{aligned} \tag{12}$$

Substituting (6), (9)–(12) into (7) completes the proof of Theorem 1. ■

The Maple program developed using the above formulas is given in Appendix for the convenience of readers.

3. Application

In this section, we present two examples to demonstrate the computational efficiency of the method developed in the previous section. We apply the obtained formulas to compute the normal forms

for these two examples, and compare the computational efficiency with existing programs. The Maple program developed in this paper (see the source code in Appendix) and the Maple program given in [Yu, 1998] are executed on a desktop machine with CPU 3.4 GHZ and 32 G RAM memory for a comparison. We have tested a number of systems and found that in general (in particular, more terms involved in the system) the method and program developed in this paper are better than the perturbation method as well as the program developed in [Yu, 1998]. Only in some special cases the situation is reversed.

In the following, for the first example, we show how to use the normal form to determine the maximum number of small amplitude limit cycles bifurcating from a focus point, as well as the maximum number of critical periods of periodic solutions in the neighborhood of the critical point. For the second example, we focus on the comparison of

computational efficiency with existing programs, and show that the recursive formulas and Maple program obtained in this paper are superior to other methods.

3.1. A five-dimensional dynamical system

The first example is a general five-dimensional dynamical system involving a number of constant parameters, given by

$$\begin{aligned} \dot{x}_1 &= a_0x_1 + x_2 + a_1x_1^3 + a_2x_1^2x_2, \\ \dot{x}_2 &= -x_1 + a_0x_2 + a_3x_2^3 \\ &\quad + a_4x_3^2 + a_5x_4x_5, \\ \dot{x}_3 &= -x_3 + a_6x_1x_2, \\ \dot{x}_4 &= -x_4 + x_5 + a_7x_1^2, \\ \dot{x}_5 &= -x_4 - x_5 + a_8x_2^2, \end{aligned} \tag{13}$$

where the $a_i, i = 0, 1, 2, \dots, 8$ are real numbers. The system has an equilibrium at the origin, and its linear part is in the Jordan canonical form, with eigenvalues, $a_0 \pm i, -1$ and $-1 \pm i$, indicating that the origin undergoes a Hopf bifurcation at the critical point $a_0 = 0$. For system (13), we compute the normal form up to 17th order, given in polar coordinates as follows:

$$\begin{aligned} \dot{r} &= r(v_0 + v_1r^2 + \dots + v_8r^{16}) + \dots, \\ \dot{\theta} &= 1 + t_0 + t_1r^2 + \dots + t_8r^{16} + \dots, \end{aligned} \tag{14}$$

where r and θ represent the amplitude and phase of motion, respectively; v_k is usually called the k th order focus value. The coefficients v_0 and t_0 are obtained from the linear analysis. For system (13), $v_0 = a_0$ and $t_0 = 0$. The first equation of (14) can be used to determine the bifurcation of limit cycles near the origin and their stability, while the second equation of (14) can be used to determine the frequency of the limit cycles. The coefficients obtained from the output of the computer program are

$$\begin{aligned} v_1 &= \frac{1}{8}a_1, \\ v_2 &= -\frac{1}{32}a_1a_2 - \frac{1}{100}a_4a_6(a_7 + 3a_8), \\ v_3 &= \frac{1}{226304000}a_1[359125a_1^2 + 5442125a_2^2 - 16a_4a_6(35509a_7 + 74168a_8)] \\ &\quad - \frac{1}{2176000}a_5[2a_5(10075a_7^4 + 11603a_7^3a_8 - 5643a_7^2a_8^2 - 4423a_7a_8^3 + 8300a_8^4) \\ &\quad - a_3a_6(7413a_7^2 + 2664a_7a_8 + 1891a_8^2)] \\ &\quad + \frac{1}{14144000}a_6[2210a_3^2a_6 - a_2a_4(105928a_7 + 22921a_8)], \\ &\quad \vdots \\ v_8 &= \frac{1474726741229822691517588834545902683}{444675434042637786245677056000000000000}a_2a_3a_4^2a_5a_6^3a_7^3a_8 + \dots, \\ t_1 &= \frac{3}{8}a_2, \\ t_2 &= -\frac{11}{256}a_1^2 - \frac{51}{256}a_2^2 - \frac{1}{160}a_4a_6(a_7 + a_8), \\ t_3 &= \frac{1}{45260800}a_1[2845375a_1a_2 - 16a_4a_6(16738a_7 + 101a_8)] \\ &\quad + \frac{1}{45260800}a_2[7839975a_2^2 - 16a_4a_6(11566a_7 - 7399a_8)] - \frac{3}{1280}a_3^2a_6^2 \end{aligned}$$

$$\begin{aligned}
 & -\frac{1}{10880000}a_5[40a_5(4800a_7^4 + 1708a_7^3a_8 - 5713a_7^2a_8^2 + 1907a_7a_8^3 - 820a_8^4) \\
 & - a_3a_6(121649a_7^2 + 9416a_7a_8 - 77889a_8^2)], \\
 & \vdots \\
 t_8 = & -\frac{112390166138349550268314971775544497279}{64033262502139841219377496064000000000000}a_2a_3a_4^2a_5a_6^3a_7^3a_8 + \dots,
 \end{aligned} \tag{15}$$

where the lengthy expressions for v_4, t_4 , etc. are omitted here for brevity.

To determine the maximum number of small amplitude limit cycles bifurcating from the origin, one may solve the polynomial equations $v_1 = v_2 = \dots = 0$ for the parameters a_i . Suppose one can obtain $v_0 = a_0 = 0, v_1(a_i) = v_2(a_i) = \dots = v_{k-1}(a_i) = 0$, but $v_k(a_i) \neq 0$, then one can conclude that at most k limit cycles may bifurcate from the origin. Moreover, by properly perturbing the parameters a_i , one can obtain k small amplitude limit cycles in the vicinity of the origin.

Now, suppose under certain conditions the origin of system (13), restricted to the center manifold, becomes a center, we can then study the critical periods of the periodical solutions around the origin. The procedure is similar to that of finding the maximum number of limit cycles, as described as follows. Let

$$\begin{aligned}
 h = r^2 > 0 \quad \text{and} \\
 p(h) = t_1h + t_2h^2 + \dots + t_kh^k + \dots
 \end{aligned} \tag{16}$$

Then, the second equation of (14) can be written as $d\theta = (1 + p(h))dt$. Let the period of motion be $T(h)$. Then, integrating this equation on both sides from 0 to 2π yields $2\pi = (1 + p(h))T(h)$, which in

turn results in

$$\begin{aligned}
 T(h) = \frac{2\pi}{1 + p(h)} \quad \text{for } 0 < h \ll 1 \\
 \text{(and so } 1 + p(h) \approx 1).
 \end{aligned} \tag{17}$$

Now, the local critical periods are determined by $T'(h) = \frac{-2\pi p'(h)}{(1+p(h))^2} = 0$. Thus, for $0 < h \ll 1$ (meaning that we consider small limit cycles), the local critical periods are determined by

$$p'(h) = t_1 + 2t_2h + \dots + kt_kh^{k-1} + \dots = 0. \tag{18}$$

Then, similar to the above discussion in determining the maximum number of limit cycles, we can find the sufficient conditions for the polynomial $p'(h)$ to have maximal number of zeros. If $t_1 = t_2 = \dots = t_{k-1} = 0$, but $t_k \neq 0$, then equation $p'(h) = 0$ can have at most $k - 1$ real roots. Hence, t_1, t_2, \dots, t_{k-1} can be perturbed to have $k - 1$ real roots, and thus system (13) can have $k - 1$ critical periods.

Next, we use (15) to determine the maximum number of small amplitude limit cycle bifurcations from the origin of system (13). To find the critical parameter values, letting $a_1 = 0$ yields $v_1 = 0$. Then, setting $a_8 = -\frac{4}{3}a_7$, we have $v_2 = 0$. With these parameter values, solving $v_3 = 0$ for a_4 yields

$$a_4 = \frac{13}{2441880a_2a_6a_7}[5508a_3^2a_6^2 + a_5a_7^2(117009a_3a_6 - 689750a_5a_7^2)], \quad (a_2a_6a_7 \neq 0).$$

Further solving $v_4 = 0$ for a_2 results in

$$a_2 = \left\{ \frac{231361[9a_3a_6(612a_3a_6 + 13001a_5a_7^2)(5508a_3^2a_6^2 + a_5a_7^2(117009a_3a_6 - 1379500a_5a_7^2)) + 475755062500a_5^4a_7^8]}{167580[299559536604a_3^2a_6^2 - a_5a_7^2(206823894568700a_5a_7^2 + 4779756148743a_3a_6)]} \right\}^{1/3},$$

and then v_5 and v_6 are simplified to

$$\begin{aligned}
 v_5 = & -\frac{481[5508a_3^2a_6^2 - a_5a_7^2(689750a_5a_7^2 - 117009a_3a_6)]}{299559536604a_3^2a_6^2 - a_5a_7^2(206823894568700a_5a_7^2 + 4779756148743a_3a_6)} \\
 & \times \{13117140119171757150379152a_3^4a_6^4 - a_5a_7^2[3964521095122924744342256808a_3^3a_6^3]
 \end{aligned}$$

$$\begin{aligned}
 & -a_5 a_7^2 (50237329268678519391542393913 a_3^2 a_6^2 + 10 a_5 a_7^2 (287295418808429080540146201219 a_3 a_6 \\
 & + 509170724293870725274253763100 a_5 a_7^2)) \Big\} \Big\}, \\
 v_6 = & -\frac{6015386 [5508 a_3^2 a_6^2 - a_5 a_7^2 (689750 a_5 a_7^2 - 117009 a_3 a_6)]^2}{343 [299559536604 a_3^2 a_6^2 - a_5 a_7^2 (206823894568700 a_5 a_7^2 + 4779756148743 a_3 a_6)]} \\
 & \times \{4369100288210241965165567410201975915584 a_3^6 a_6^6 \\
 & + a_5 a_7^2 [48714467325624956765186957094776399777904 a_3^5 a_6^5 \\
 & - a_5 a_7^2 (3845973325800546252799748753542377356118852 a_3^4 a_6^4 \\
 & + a_5 a_7^2 [103213562801159168000556264482023982522738583 a_3^3 a_6^3 \\
 & + 4 a_5 a_7^2 (197025074518409040487257210142206321678001863 a_3^2 a_6^2 \\
 & - 25 a_5 a_7^2 (16558651465224677787932424690182456471021097 a_3 a_6 \\
 & + 196414263618794085309604077817625129955519500 a_5 a_7^2))] \Big\} \Big\}.
 \end{aligned}$$

It can be shown that besides the common factor in v_5 and v_6 , the only possible parameter values for $v_5 = v_6 = 0$ are $a_5 a_7 = a_3 a_6 = 0$, which are obviously not allowed. This suggests that there exist parameter values such that $v_i = 0, i = 0, 1, \dots, 5$, but $v_6 \neq 0$. Therefore, for system (13), we can at most have six small amplitude limit cycles bifurcating from the origin. To find the parameter values such that $v_5 = 0$, we may set $a_5 = a_6 = a_7 = 1$ and then solve an equation from $v_5 = 0$ for a_3 , yielding four real solutions. Choosing one of them, we have the following set of critical values:

$$\begin{aligned}
 a_5 &= a_6 = a_7 = 1, \\
 a_0 &= a_1 = 0, \\
 a_8 &= -\frac{4}{3}, \\
 a_3 &= -1.8401519905 \dots, \\
 a_2 &= -0.1765825495 \dots, \\
 a_4 &= 26.7243747795 \dots,
 \end{aligned}$$

under which $v_i = 0, i = 0, 1, \dots, 5, v_6 = 0.0573817846 \dots$. Then making proper perturbations in backward order, on a_3 for v_5 , on a_2 for v_4 , on a_4 for v_3 , on a_8 for v_2 , and then on a_1 for v_1 , and finally on a_0 for v_0 such that

$$0 < v_0 \ll -v_1 \ll v_2 \ll -v_3 \ll v_4 \ll -v_5 \ll v_6,$$

leading to six small limit cycles.

Next, we consider critical periods of periodic solutions near the origin. To do this, we first need

to find the conditions under which the origin is a center, restricted to the center manifold. There are a number of such conditions. Here, we consider one satisfying

$$a_1 = a_5 = a_6 = 0, \tag{19}$$

under which

$$\begin{aligned}
 t_1 &= \frac{3}{8} a_2, \quad t_2 = -\frac{51}{256} a_2^2, \quad t_3 = \frac{1419}{8192} a_2^3, \\
 t_4 &= -\frac{47505}{262144} a_2^4, \quad t_5 = \frac{438825}{2097152} a_2^5, \quad \dots
 \end{aligned}$$

Therefore, under the condition (19), system (13) does not have critical periods near the origin; it is either monotonically increasing for $a_2 > 0$ or decreasing for $a_2 < 0$. When $a_2 = 0$, the origin is a isochronous center.

3.2. A three-dimensional competitive Lotka–Volterra system

In this section, we consider a three-dimensional competitive Lotka–Volterra system, described by the following differential equations:

$$\dot{x}_i = x_i \left(r_i - \sum_{j=1}^3 a_{ij} x_j \right), \quad i = 1, 2, 3, \tag{20}$$

where x_i represents the population of i th species, and the coefficients take positive real values, $r_i > 0$,

$a_{ij} > 0, i, j = 1, 2, 3$. Over the last twenty years, a number of articles concerning the bifurcation of limit cycles for system (20) have been published (e.g. see [Hofbauer & So, 1994; Lu & Luo, 2002; Gyllenberg *et al.*, 2006; Gyllenberg & Yan, 2009]). Particularly, for system (20) four limit cycles were found by Gyllenberg and Yan [2009], using appropriate parameter values. These four limit cycles include three small amplitude limit cycles, proved by using focus value computation, and one large limit cycle, shown by constructing a heteroclinic cycle. In this section, we consider the Hopf bifurcation emerging from an interior singular point and use the normal form (or focus values) to study the maximum number of limit cycles bifurcating from this point.

It is noted that system (20) has a total of 12 parameters. Since we are interested in the limit cycles bifurcating from an interior equilibrium solution of system (20), we, without loss of generality, may assume that $E = (1, 1, 1)$ is the equilibrium solution, which yields $r_i = \sum_{j=1}^3 a_{ij}$ and reduce the number of parameters to nine. Taking the transla-

tion $x_i \rightarrow x_i + 1$ such that the equilibrium solution is moved to the origin, we have

$$\dot{x}_i = -(1 + x_i) \left(\sum_{j=1}^3 a_{ij} x_j \right). \quad (21)$$

The Jacobian of system (21) at $x = 0$ is the matrix $A = (-a_{ij})$, which has the characteristic polynomial $P(\lambda) = \lambda^3 - T\lambda^2 + M\lambda - D$, where

$$\begin{aligned} T &= -(a_{11} + a_{22} + a_{33}), \\ M &= a_{11}a_{22} - a_{12}a_{21} + a_{11}a_{33} - a_{13}a_{31} \\ &\quad + a_{22}a_{33} - a_{23}a_{32}, \\ D &= \det(A). \end{aligned} \quad (22)$$

When $TM = D$ and $M > 0$, there exist a pair of purely imaginary eigenvalues $\pm i\sqrt{M}$ and a negative eigenvalue T , and Hopf bifurcation occurs.

A parameter, say a_{31} , is needed to satisfy $TM = D$. Moreover, one may apply a time scaling to set $M = 1$, using a parameter, say a_{32} . Finally, we may use a parameterization so that $a_{33} = 1$. Solving equations $TM = D$ and $M = 1$ yields

$$\begin{aligned} a_{31} &= \frac{a_{11}^2 a_{22} a_{23} + a_{11}^2 a_{23} - a_{11} a_{13} a_{22} a_{21} - a_{11} a_{21} a_{13} - a_{11} a_{23} a_{21} a_{12} \\ &\quad + a_{13} a_{21}^2 a_{12} - a_{13} a_{22} a_{21} + a_{21} a_{13} + a_{22} a_{23} + a_{23} + a_{23} a_{21} a_{12}}{a_{23}^2 a_{12} - a_{13}^2 a_{21} - a_{13} a_{22} a_{23} + a_{23} a_{11} a_{13}}, \\ a_{32} &= \frac{-a_{13} - a_{11} a_{13} - a_{12} a_{23} + a_{11} a_{22} a_{12} a_{23} - a_{21} a_{12}^2 a_{23} - a_{13} a_{22}^2 - a_{13} a_{11} a_{22}^2 \\ &\quad - a_{13} a_{21} a_{12} + a_{22} a_{12} a_{23} + a_{11} a_{12} a_{23} + a_{22} a_{21} a_{12} a_{13}}{a_{23}^2 a_{12} - a_{13}^2 a_{21} - a_{13} a_{22} a_{23} + a_{23} a_{11} a_{13}}. \end{aligned}$$

Thus, only six parameters $a_{ij}, i = 1, 2, j = 1, 2, 3$ are left for determining the focus values. In general, we might be able to find the parameter values satisfying $v_1 = v_2 = \dots = v_6 = 0$, but $v_7 \neq 0$, and thus seven small limit cycles may be found in the vicinity of the equilibrium solution. If, in addition, there exists a large limit cycle near the heteroclinic loop, then the maximum number of limit cycles becomes eight.

To apply the Maple program, we first need to put the linear part of system (20) in Jordan canonical form. To achieve this, introducing the linear variable transformation $x \rightarrow \mathfrak{X}x$, where

$$\mathfrak{X} = \begin{pmatrix} a_{13}a_{22} - a_{12}a_{23} & a_{13} & a_{13}(a_{11} + 1) + a_{12}a_{23} \\ a_{11}a_{23} - a_{13}a_{21} & a_{23} & a_{23}(a_{22} + 1) + a_{13}a_{21} \\ a_{12}a_{21} - a_{11}a_{22} + 1 & -(a_{11} + a_{22}) & (a_{11} + 1)(a_{22} + 1) - a_{12}a_{21} \end{pmatrix}$$

into system (21) yields the following system:

$$\dot{x}_1 = x_2 + q_1(x), \quad \dot{x}_2 = -x_1 + q_2(x), \quad \dot{x}_3 = Tx_3 + q_3(x), \quad (23)$$

where $T = -(a_{11} + a_{22} + 1)$ and $q_i(x), i = 1, 2, 3$ are quadratic homogenous polynomials, given in the form as

$$q_i = b_{i200}x_1^2 - b_{i200}x_2^2 + b_{i002}x_3^2 + b_{i110}x_1x_2 + b_{i102}x_1x_3 + b_{i011}x_2x_3, \quad i = 1, 2, 3,$$

where the coefficients b_{ijkl} 's are lengthy expressions in terms of the original six coefficients a_{ij} , $i = 1, 2$; $j = 1, 2, 3$. We have executed the Maple program developed in this paper and that given in [Yu, 1998] on the desktop machine to obtain the following results. For the Maple program given in [Yu, 1998], it took 251 min CPU time and 12.35 GB Ram memory to get the focus values up to third order; while for the program developed in this paper, it only took 31 min CPU time and 3.86 GB Ram memory to get the focus values up to fourth order. This clearly shows that the recursive formulas derived and Maple program developed in this paper are more computationally efficient for higher-order normal forms than that given in [Yu, 1998], though the program in [Yu, 1998] was proved computationally efficient, in particular, for lower-order normal forms. In order to get higher-order focus values of system (23), we need a more powerful machine with higher memory.

The first focus value obtained from the computer output is

$$v_1 = \frac{1}{8(4 + T^2)} \{ 2[(b_{2011} - b_{1101})T - 2(b_{1011} + b_{2101})]b_{3200} - [(b_{1011} + b_{2101})T + 2(b_{2011} - b_{1101})]b_{3110} \}.$$

The focus values starting from the second one have very long expressions. The number of terms in each of the focus values are given below:

Focus Value	Number of Terms
v_1	8
v_2	1036
v_3	24088
v_4	261401

It can be seen that the number of terms increase very rapidly as the order of the focus values increase. Moreover, when the original parameters a_{ij} are substituted into these expressions, they even have more terms. Thus, finding possible values of the six parameters $a_{ij} > 0$, $i = 1, 2$; $j = 1, 2, 3$ such that $v_j = 0$, $j = 1, 2, \dots, 6$, but $v_7 \neq 0$, is very difficult and challenging. It not only needs power computer systems (high speed with large memory), but also needs efficient polynomial solvers implemented with a computer algebra system such as Maple.

4. Conclusion

In this paper, we have derived explicit recursive formulas for computing normal forms and center manifold of general n -dimensional dynamical systems associated with Hopf bifurcation. Maple program has also been developed, which is convenient in application. Two examples are presented to show that the method and program developed in this paper are computationally efficient.

References

- Algaba, A., Freire, E. & Gamero, E. [1998] "Hypernormal form for the Hopf-zero bifurcation," *Int. J. Bifurcation and Chaos* **8**, 1857–1887.
- Bautin, N. [1952] "On the number of limit cycles appearing from an equilibrium point of the focus or center type under varying coefficients," *Matem. Sb.* **30**, 181–196.
- Chow, S.-N., Li, C.-Z. & Wang, D. [1994] *Normal Forms and Bifurcations of Planar Vector Fields* (Cambridge University Press, Cambridge).
- Gazor, M. & Yu, P. [2012] "Spectral sequences and parametric normal forms," *J. Diff. Eqs.* **252**, 1003–1031.
- Giné, J. & Santallusia, X. [2001] "On the Poincaré–Lyapunov constants and the Poincaré series," *Appl. Math. (Warsaw)* **28**, 17–30.
- Giné, J. & Santallusia, X. [2004] "Implementation of a new algorithm of computation of the Poincaré–Lyapunov constants," *J. Comput. Appl. Math.* **166**, 465–476.
- Giné, J. & Mallol, J. [2009] "Minimum number of ideal generators for a linear center perturbed by homogeneous polynomials," *Nonlin. Anal.* **71**, 132–137.
- Guckenheimer, J. & Holmes, P. [1993] *Nonlinear Oscillations, Dynamical Systems, and Bifurcations of Vector Fields*, 4th edition (Springer-Verlag, NY).
- Gyllenberg, M., Yan, P. & Wang, Y. [2006] "A 3-D competitive Lotka–Volterra system with three limit cycles: A falsification of a conjecture by Hofbauer and So," *Appl. Math. Lett.* **19**, 1–7.
- Gyllenberg, M. & Yan, P. [2009] "Four limit cycles for a three-dimensional competitive Lotka–Volterra system with a heteroclinic cycle," *Comput. Math. Appl.* **58**, 649–669.
- Hofbauer, J. & So, J. W.-H. [1994] "Multiple limit cycles for three dimensional Lotka–Volterra equations," *Appl. Math. Lett.* **7**, 65–70.
- Li, J. [2003] "Hilbert's 16th problem and bifurcations of planar polynomial vector fields," *Int. J. Bifurcation and Chaos* **13**, 47–106.
- Lloyd, N. G. & Pearson, J. M. [1990] "REDUCE and the bifurcation of limit cycles," *J. Symb. Comput.* **9**, 215–224.

- Lu, Z. & Luo, Y. [2002] “Two limit cycles in three-dimensional Lotka–Volterra systems,” *Comput. Math. Appl.* **44**, 51–66.
- Nayfeh, A. H. [1993] *The Methods of Normal Forms* (John Wiley, NY).
- Yu, P. [1998] “Computation of the normal forms via a perturbation technique,” *J. Sound Vibr.* **211**, 19–38.
- Yu, P. [1999] “Simplest normal forms of Hopf and generalized Hopf bifurcations,” *Int. J. Bifurcation and Chaos* **9**, 1917–1939.
- Yu, P. [2001] “Symbolic computation of normal forms for resonant double Hopf bifurcations using multiple time scales,” *J. Sound Vibr.* **247**, 615–632.
- Yu, P. [2002] “Analysis on double Hopf bifurcation using computer algebra with the aid of multiple scales,” *Nonlin. Dyn.* **27**, 19–53.
- Yu, P. [2003] “A simple and efficient method for computing center manifold and normal forms associated with semisimple cases,” *Dyn. Contin. Discr. Impul. Syst. Series B: Appl. Algorithm.* **10**, 273–286.
- Yu, P. & Yuan, Y. [2003] “An efficient method for computing the simplest normal forms of vector fields,” *Int. J. Bifurcation and Chaos* **13**, 19–46.
- Yu, P. & Leung, A. Y. T. [2003a] “The simplest normal form of Hopf bifurcation,” *Nonlinearity* **16**, 277–300.
- Yu, P. & Leung, A. Y. T. [2003b] “A perturbation method for computing the simplest normal forms of dynamical systems,” *J. Sound Vibr.* **261**, 123–151.

Appendix

This symbolic Maple program is developed on the basis of the formulas in Theorem 1, which can be used to find the normal forms of Hopf bifurcations of general n -dimensional systems. Here, the input is for the second example in the section of application.

```
with(LinearAlgebra):
M1 := 1: # No. of nonzero real eigenvalues
M2 := 0: # No. of complex conjugate eigenvalues
N := 2: # the highest degree of the vector field
Ord := 13:
M := 2 + M1 + M2*2:
L := 1 + M1 + M2:
f[1] := x[2]+b1200*x[1]^2-b1200*x[2]^2+b1002*x[3]^2
      +b1110*x[1]*x[2]+b1101*x[1]*x[3]+b1011*x[2]*x[3]:
f[2] := -x[1]+b2200*x[1]^2-b2200*x[2]^2+b2002*x[3]^2
      +b2110*x[1]*x[2]+b2101*x[1]*x[3]+b2011*x[2]*x[3]:
f[3] := -alpha*x[3]+b3200*x[1]^2-b3200*x[2]^2+b3002*x[3]^2
      +b3110*x[1]*x[2]+b3101*x[1]*x[3]+b3011*x[2]*x[3]:
x[1] := v[1]+v[L+1]:
x[2] := I*(v[1]-v[L+1]):
f[1] := simplify(f[1]-I*f[2])/2:
IEf[1] := diff(f[1],v[1]):
j := 3:
for n from 2 to M1+1 do
  x[j] := v[n]:
  f[n] := simplify(f[j]):
  IEf[n] := diff(f[n],v[n]):
  j := j+1:
od:
k := L+2:
for n from M1+2 to L do
  x[j] := v[n]+v[k]:
  x[j+1] := I*(v[n]-v[k]):
  f[n] := simplify(f[j]-I*f[j+1])/2:
  IEf[n] := diff(f[n],v[n]):
  j := j+2:
  k := k+1:
od:
for j to L do
```

```

for k from 1 to M do
  IEf[j] := subs(v[k]=0,IEf[j]):
od:
REf[j] := subs(I=0,IEf[j]):
IEf[j] := (IEf[j]-REf[j])/I:
od:
SizeIndex := Array(1..L,2..N):
Mr := seq(1,i=1..M):
for m from 2 to N do
  i := 1:
  Mv := [seq(0,j=1..M)]:
  temp := 1:
  Mv[1]:= m+1:
  Mv[2]:= -1:
  while i < M do
    Mv[i+1] := 1+Mv[i+1]:
    Mv[1] := Mv[i]-1:
    if i<>1 then Mv[i]:=0: fi:
    if Mv[i+1]=1 then temp:=temp+1: fi:
    if Mv[1]=0 then temp:=temp-1: i := i+1:
    else i:=1: fi:
    Mcv := [[seq(0,j=1..temp+2)], [seq(0,j=1..temp+2)]]:
    Mcv[1,1] := temp:
    Mcv[2,1] := m+add(Mv[n],n=2..L)+add(Mv[n],n=L+2..M):
    j := 2:
    for k from i to M do
      if Mv[k]<>0 then
        Mcv[1,j] := k:
        Mcv[2,j] := Mv[k]:
        j := j+1:
      fi:
    od:
    for j from 1 to L do
      coef := f[j]:
      for k to M do
        if Mv[k]=0 then
          coef := subs(v[k]=0,coef):
        else
          coef := subs(v[k]=0,diff(coef,'$(v[k],Mv[k])))/factorial(Mv[k]):
        fi:
      od:
      if coef<>0 then
        SizeIndex[j,m] := SizeIndex[j,m]+1:
        Mcv[1,-1] := subs(I=0,coef):
        Mcv[2,-1] := subs(I=1,coef-Mcv[1,-1]):
        Index[j,m,SizeIndex[j,m]] := Mcv:
        Mr:=seq(max(Mr[n],Mv[n]),n=1..M):
      fi:
    od:
  od:
  Mr := [max(Mr[1],Mr[1+L]),seq(Mr[n],n=2..M1+1),seq(max(Mr[n],Mr[n+M2+1]),n=M1+2..L)]:
  Rh[1,1,1,1] := 1:
  Ih[1,1,1,1] := 0:
  Rh[1,1,1,0] := 0:

```



```

Ih[1,1,1,0] := 0:
si := 1/2:
for s from 2 to Ord do
print('order =',s):
  si := si+1/2:
  Ms := min(Mr[1],s):
  for k from 2 to Ms do
    for i from 0 to s do
      Ih[1,k,s,i] := 0:
      Rh[1,k,s,i] := 0:
      for m from k-1 to s-1 do
        for l from max(0,i+m-s) to min(m,i) do
          Rh[1,k,s,i] := Rh[1,k,s,i] + Rh[1,k-1,m,l]*Rh[1,1,s-m,i-1]
            - Ih[1,k-1,m,l]*Ih[1,1,s-m,i-1]:
          Ih[1,k,s,i] := Ih[1,k,s,i] + Ih[1,k-1,m,l]*Rh[1,1,s-m,i-1]
            + Rh[1,k-1,m,l]*Ih[1,1,s-m,i-1]:
        od:
      od:
    od:
  for j from 2 to L do
    Ms := min(Mr[j],si):
    for k from 2 to Ms do
      for i from 0 to s do
        Ih[j,k,s,i] := 0:
        Rh[j,k,s,i] := 0:
        for m from 2*k-2 to s-2 do
          for l from max(0,i+m-s) to min(m,i) do
            Rh[j,k,s,i] := Rh[j,k,s,i] + Rh[j,k-1,m,l]*Rh[j,1,s-m,i-1]
              - Ih[j,k-1,m,l]*Ih[j,1,s-m,i-1]:
            Ih[j,k,s,i] := Ih[j,k,s,i] + Ih[j,k-1,m,l]*Rh[j,1,s-m,i-1]
              + Rh[j,k-1,m,l]*Ih[j,1,s-m,i-1]:
          od:
        od:
      od:
    od:
  for j from 1 to L do
    for k from 0 to si do
      if s=Ord then k:=iquo(s+1,2): fi:
      if k<si then nk:=2: else nk := 1: fi:
      temp := min(s-k, k, si-1):
      sk := k:
      for t from 1 to nk do
        Ra[t] := 0:
        Ia[t] := 0:
        for m from 1 to temp do
          Ra[t] := Ra[t] - (s-2*m)*Ren[m]*Rh[j,1,s-2*m,sk-m]
            + (2*sk-s)*Imn[m]*Ih[j,1,s-2*m,sk-m]:
          Ia[t] := Ia[t] - (2*sk-s)*Imn[m]*Rh[j,1,s-2*m,sk-m]
            - (s-2*m)*Ren[m]*Ih[j,1,s-2*m,sk-m]:
        od:
      sk := s-sk:
    od:
  for m from 2 to min(s,N) do

```



```

Size := SizeIndex[j,m]:
for i from 1 to Size do
  Mv := Index[j,m,i]:
  Nonzero := Mv[1,1]:
  Sleft := s-Mv[2,1]:
  if Sleft>=0 then
    NIs:= binomial(Sleft+Nonzero-1,Sleft):
    S := Vector(Nonzero):
    for l to NIs do
      if l=1 then
        S[1]:= Sleft:
        p := 1:
      else
        S[p+1]:= S[p+1]+1:
        S[1] := S[p]-1:
        if p <> 1 then
          S[p] := 0:
        fi:
        if S[1]=0 then
          p := p+1:
        else
          p := 1:
        fi:
      fi:
    for r from 1 to Nonzero do
      if Mv[1,r+1]=1 or Mv[1,r+1]=L+1 then
        Sv[r] := S[r]+Mv[2,r+1]:
      else
        Sv[r] := S[r]+2*Mv[2,r+1]:
      fi:
    od:
    Sv[r]:= k:
    Ks := k+1:
    q := Nonzero:
    Kv[q]:= -1:
    while q <= Nonzero do
      Kv[q]:= Kv[q]+1:
      Ks := Ks-1:
      temp := Ks:
      qq := 1:
      while temp >= Sv[qq] do
        Kv[qq] := Sv[qq]:
        temp := temp-Sv[qq]:
        qq := qq+1:
      od:
      Kv[qq] := temp:
      for n from qq+1 to q-1 do
        Kv[n] := 0:
      od:
      for t from 1 to nk do
        Rei := Mv[1,-1]:
        Imi := Mv[2,-1]:
        for n from 2 to Nonzero+1 do
          teR := Rei:
          if Mv[1,n]<L+1 then

```

```

Rei := teR*Rh[Mv[1,n],Mv[2,n],Sv[n-1],Kv[n-1]]
      -Imi*Ih[Mv[1,n],Mv[2,n],Sv[n-1],Kv[n-1]]:
Imi := teR*Ih[Mv[1,n],Mv[2,n],Sv[n-1],Kv[n-1]]
      +Imi*Rh[Mv[1,n],Mv[2,n],Sv[n-1],Kv[n-1]]:
Kv[n-1] := Sv[n-1]-Kv[n-1]:
else
Kv[n-1] := Sv[n-1]-Kv[n-1]:
if Mv[1,n]=L+1 then
  T := 1:
else
  T := Mv[1,n]-M2-1:
fi:
Rei := teR*Rh[T,Mv[2,n],Sv[n-1],Kv[n-1]]
      +Imi*Ih[T,Mv[2,n],Sv[n-1],Kv[n-1]]:
Imi := Imi*Rh[T,Mv[2,n],Sv[n-1],Kv[n-1]]
      -teR*Ih[T,Mv[2,n],Sv[n-1],Kv[n-1]]:
fi:
od:
Ra[t] := Ra[t]+Rei:
Ia[t] := Ia[t]+Imi:
od:
if t=2 then
  for n from 1 to Nonzero do
    Kv[n] := Sv[n]-Kv[n]:
  od:
fi:
if Nonzero=1 or k = 0 then
  break:
fi:
if qq>1 then
  q := qq:
  Ks:= Ks-temp:
else
  if Kv[1]=0 then
    Ks:= Kv[q]:
    q := q+1:
  else
    Ks:= Kv[1]:
    q := 2:
  fi:
  while Sv[q]=Kv[q] do
    Ks:= Ks+Kv[q]:
    q := q+1:
  od:
fi:
od:
od:
fi:
od:
for t from 1 to nk do
  Ra[t] := factor(Ra[t]):
  Ia[t] := factor(Ia[t]):
od:
od:
for t from 1 to nk do

```

```

    if j=1 then
      if k=(s+1)/2 then
        Ren[k-1] := Ra[t]:
        Imn[k-1] := Ia[t]:
        Rh[1,1,s,k] := 0:
        Ih[1,1,s,k] := 0:
      else
        Rh[1,1,s,k] := Ia[t]/(2*k-s-1):
        Ih[1,1,s,k] := -Ra[t]/(2*k-s-1):
      fi:
    else
      temp := 2*k-s-IEf[j]:
      Rh[j,1,s,k] := (-REf[j]*Ra[t]+temp*Ia[t])/(REf[j]*REf[j]+temp*temp):
      Ih[j,1,s,k] := (-REf[j]*Ia[t]-temp*Ra[t])/(REf[j]*REf[j]+temp*temp):
    fi:
    k := s-k:
  od:
od:
if s=Ord then
  break:
fi:
od:
od:
save Ren, Imn, 'output':
```